
1 Overview

This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 8.6.1.3 which is part of the ‘iwd-TTCN3-B2018-03_D18wk37’ ATS delivery. 

With these changes applied the test case can be demonstrated to run with one or more UEs (see section 5). Execution log files are provided as evidence.

3 Verification Test Summary

Test Case:
TC_8_6_1_3
Test Group:
                       LTE_A_PRO\8_6\RRC_Immediate_MDT_PRO.ttcn
ATS Version:
iwd-TTCN3-B2018-03_D18wk37
System Simulator used:
Anite Conformance Toolset Solution

UE used:
                       INTEL 7660
Verification Status:          
PASS

4 Corrections required for test case 8.6.1.3
Change 1
	Function name
	function f_TC_8_6_1_3_EUTRA()

	Reason for change
	1. Function f_EUTRA_Get_508_TrafficFlowTemplate() should be called only after the EUTRA PTC has been initialized (due to the embedded call to f_EUTRA_MobileInfo_GetAssignedPdnType()).
2. The UE may stop sending Scheduling Requests before the timer t_ReportInterval has expired at step 9 and 16 because the default value of Dsr_TransMax is 4. 
3.  MeasObjectToAddModList and MeasIdToAddModList are missing in the mesConfig at step 1.
4.  At step6, the check needs to be repeated until timer t_ReportInterval has expired or a measurement report is received containing measResultCSI_RS_List_r12.
5.  At step 9 and 15, the f_EUTRA_StopULGrantTransmission() will stop the sending of TA from the SS which may cause the UE to lose UL synchronisation.
6. At step 9 and 15, UL grant needs to be stopped and then restarted after the duration indicated in IE reportInterval.
7. The data packets sent at steps 3 and 4 are not valid IP packets (source and destination addresses are not populated)


	Summary of change
	1. Function calls to f_EUTRA_Get_508_TrafficFlowTemplate() has been moved after f_EUTRA_Init().
2. The value of Dsr_TransMax has been configured as 64 so that UE can send maximum number of SR when the UL grant is stopped st step 9 and 16.
3. Changes done in RRC Reconfiguraiton->MeasConfig to add MeasObjectToAddModList andMeasIdToAddModList
4. Add repeat statements at step 6 in the case that a MeasurementReport is received that does not contain measResultCSI_RS_List_r12, or any DRB data is received.
5. f_EUTRA_StopULGrantTransmission_PUCCH_Sync should be used instead of f_EUTRA_StopULGrantTransmission so that SS will still keep sending TA even after uplink grant is stoped.
6. Delay of 750ms has been added before steps 10 And 16 . After the delay uplink grants are restarted.
7. Valid IP packet created for DRB associated with EPS bearer context 1 and 3

Note : A prose CR on 36.523-1 associated with changes 2, 3 and 6 will be raised at RAN5#81.



	TTCN module
	RRC_Immediate_MDT_PRO

	MCC160 Comment
	


Before change

   function f_TC_8_6_1_3_EUTRA() runs on EUTRA_PTC

  { // Immediate MDT / Measurement / Latency metrics for UL PDCP Packet Delay per QCI

    const integer tsc_IdReportConfig_PeriodicalCsiRs := 1;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var integer v_EpsQosBitrateForVideo := 2104;        // Bandwidth requirement of the H.264 codec to EPS bearer mappings as specified in Appendix E.36 of TS 26.114 sic@

    var B8_Type v_EncodeBitRate := f_EPS_QualityOfService_EncodeBitRate(v_EpsQosBitrateForVideo);

    var template (value) EPS_QoSList_Type v_Qos_List :={cs_508_EPS_QoS_Dedicated_1, cds_EPS_QoS_Def_nonGBR_QCI, cds_EPS_QoS_Dedicated_3(v_EncodeBitRate)};

    var template (value) TrafficFlowTemplate v_Tft_Dedicated1 := f_EUTRA_Get_508_TrafficFlowTemplate(1, tsc_EpsDedicatedBearerId);   // Dedicated EPS bearer context #1

    var template (value) TrafficFlowTemplate v_Tft_Default2 := f_EUTRA_Get_508_TrafficFlowTemplate(2, tsc_EpsDefaultBearerId2);      // Default EPS bearer #2

    var template (value) TrafficFlowTemplate v_Tft_Dedicated3 := f_EUTRA_Get_508_TrafficFlowTemplate(3, tsc_EpsDedicatedBearerId3);  // Dedicated EPS bearer context #3

    var template (value) Tft_List_Type v_Tft_List := {v_Tft_Dedicated1, v_Tft_Default2, v_Tft_Dedicated3};

    var PDCP_SDUList_Type v_PDCP_SDUList := { f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE()) };      //IP packet to transmit
    var RadioBearerId_Type v_RbId_DedicatedEPS1  := {Drb:=f_EUTRA_NB_EpsBearerAssociatedDRB(tsc_EpsDedicatedBearerId)};

    var RadioBearerId_Type v_RbId_DedicatedEPS3  := {Drb:=f_EUTRA_NB_EpsBearerAssociatedDRB(tsc_EpsDedicatedBearerId3)};

    var SRB_COMMON_IND v_MeasReport;

    var template UL_PDCP_DelayResultList_r13 v_MeasList := {permutation(cr_UL_PDCP_DelayResult(qci1), cr_UL_PDCP_DelayResult(qci2))}

    timer t_ReportInterval := 5.120; // 5120ms
    // Init variables

    f_EUTRA_Init(c1);

    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -70);

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    // The UE is in state Registered, Idle Mode (State 2) according to [18].

    f_EUTRA_InitialRegistration(eutra_Cell1, PREAMBLE, STATE2A_TESTLOOP_ModeB, NORMAL);

    f_EUTRA_TestBody_Set (true);

    //@siclog "Step 0" siclog@

    /* The SS performs the generic procedure to get UE in Loopback Activated (state 4) with condition UE TEST LOOP MODE B with a Reference default EPS bearer context #1 and

    Reference dedicated EPS bearer context #1 and Reference default EPS bearer #2 and Reference dedicated EPS bearer context #3 according to [18] on Cell A */

    f_EUTRA_GenericRbEst(eutra_CellA, 3, 0, v_Qos_List, v_Tft_List);

    f_EUTRA_CloseUE_TestLoopModeB(eutra_Cell1, '00'O);

    //@siclog "Step 1-2" siclog@

    //The SS transmits an  RRCConnectionReconfiguration message including MeasConfig to setup UL-DelayConfig

   f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(v_RRC_TI,

                                                                                                     cs_MeasConfig(omit,

                                                                                                                   omit,

                                                                                                                   omit,

                                                                                                                   cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_PeriodicalCsiRs,

                                                                                                                                                      cs_ReportConfigToAddMod_ReportConfigEUTRA_MeasCSI_RS_Id(ms5120, r64, cs_UL_DelayConfig_Setup_r13)),

                                                                                                                   omit,

                                                                                                                   omit)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 2");

    //@siclog "Step 3" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #1

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS1, cs_TimingInfo_Now, v_PDCP_SDUList));
    //@siclog "Step 4" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #3

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS3, cs_TimingInfo_Now, v_PDCP_SDUList));
    //@siclog "Step 5" siclog@

    //SS provides sufficient UL grants to ensure UE can loopback IP packets without delay and waits for duration indicated in the IE reportInterval

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);

    //@siclog "Step 6" siclog@

    //Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2

    t_ReportInterval.start;

    alt{

        [] SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MeasurementReport_CheckAny))-> value v_MeasReport

        {

            if(ispresent(v_MeasReport.Signalling.Rrc.Dcch.message_.c1.measurementReport.criticalExtensions.c1.measurementReport_r8.measResults.measResultCSI_RS_List_r12))

            {

                t_ReportInterval.stop;

                f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 6: UE transmits a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2");

            }

        }

        [] t_ReportInterval.timeout

        {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 6: UE does not transmit a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2");

        }

    }

    //@siclog "Step 7" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #1

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS1, cs_TimingInfo_Now, v_PDCP_SDUList));
    //@siclog "Step 8" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #3

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS3, cs_TimingInfo_Now, v_PDCP_SDUList));
    //@siclog "Step 9" siclog@

    //SS doesn't provides UL grants to UE and waits for duration indicated in the IE reportInterval

    f_EUTRA_StopULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now);
    //@siclog "Step 10" siclog@

    //Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2

    t_ReportInterval.start;

    alt{

        [] SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cdr_MeasurementReport_UL_PDPC_DelayResults(1,-,-,-,-,v_MeasList)))

        {

            t_ReportInterval.stop;

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 10: UE transmits a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and QCI 2");

        }

        [] t_ReportInterval.timeout

        {

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 10: UE transmits a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2");

        }

    }

    //@siclog "Step 11-12" siclog@

    //The SS transmits an RRCConnectionReconfiguration message including MeasConfig to release the UL-DelayConfig measurement

    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(v_RRC_TI,

                                                                                                     cs_MeasConfig(omit,

                                                                                                                   omit,

                                                                                                                   omit,

                                                                                                                   cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_PeriodicalCsiRs,

                                                                                                                                                      cs_ReportConfigToAddMod_ReportConfigEUTRA_MeasCSI_RS_Id(ms10240, r1, cs_UL_DelayConfig_Release_r13)),

                                                                                                                   omit,

                                                                                                                   omit)));

    //@siclog "Step 13" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #1

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS1, cs_TimingInfo_Now, v_PDCP_SDUList));

    //@siclog "Step 14" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #3

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS3, cs_TimingInfo_Now, v_PDCP_SDUList));

    //@siclog "Step 15" siclog@

    //SS doesn't provides UL grants to UE and waits for duration indicated in the IE reportInterval

    //@siclog "Step 16" siclog@

    //Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and QCI 2

    t_ReportInterval.start;

    alt{

        [] SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MeasurementReport_CheckAny))-> value v_MeasReport

        {

            if(ispresent(v_MeasReport.Signalling.Rrc.Dcch.message_.c1.measurementReport.criticalExtensions.c1.measurementReport_r8.measResults.measResultCSI_RS_List_r12))

            {

                t_ReportInterval.stop;

                f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 16: UE transmits a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2");

            }

        }

        [] t_ReportInterval.timeout

        {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 16: UE does not transmit a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2");

        }

    }

    f_EUTRA_TestBody_Set (false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);

  }
After change

  import from EUTRA_BandDependentParam all; 

  import from EUTRA_AuxiliaryFunctions all;

  import from EUTRA_L2_CommonFunctions all;

  import from EUTRA_CellCfg_Templates all;

  import from EUTRA_MAC_Common all;
function f_TC_8_6_1_3_EUTRA() runs on EUTRA_PTC

  { // Immediate MDT / Measurement / Latency metrics for UL PDCP Packet Delay per QCI

    const integer tsc_IdReportConfig_PeriodicalCsiRs := 1;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var integer v_EpsQosBitrateForVideo := 2104;        // Bandwidth requirement of the H.264 codec to EPS bearer mappings as specified in Appendix E.36 of TS 26.114 sic@

    var B8_Type v_EncodeBitRate := f_EPS_QualityOfService_EncodeBitRate(v_EpsQosBitrateForVideo);

    var template (value) EPS_QoSList_Type v_Qos_List :={cs_508_EPS_QoS_Dedicated_1, cds_EPS_QoS_Def_nonGBR_QCI, cds_EPS_QoS_Dedicated_3(v_EncodeBitRate)};

    var template (value) TrafficFlowTemplate v_Tft_Dedicated1;   // Dedicated EPS bearer context #1

    var template (value) TrafficFlowTemplate v_Tft_Default2;      // Default EPS bearer #2

    var template (value) TrafficFlowTemplate v_Tft_Dedicated3;  // Dedicated EPS bearer context #3

    var template (value) Tft_List_Type v_Tft_List;    

    var charstring v_IP_Address1_NW;

    var B3_Type v_PdnType;

    var boolean v_UseIPv4;    

    var octetstring v_UdpPacket_DRB1;

    var octetstring v_UdpPacket_DRB3;

    var SR_PUCCH_ResourceIndex_Type v_Sr_PUCCH_ResourceIndex;

    var SR_ConfigIndex_Type v_SR_ConfigIndex;
    var RadioBearerId_Type v_RbId_DedicatedEPS1  := {Drb:=f_EUTRA_NB_EpsBearerAssociatedDRB(tsc_EpsDedicatedBearerId)};

    var RadioBearerId_Type v_RbId_DedicatedEPS3  := {Drb:=f_EUTRA_NB_EpsBearerAssociatedDRB(tsc_EpsDedicatedBearerId3)};

    var SRB_COMMON_IND v_MeasReport;

    var template UL_PDCP_DelayResultList_r13 v_MeasList := {permutation(cr_UL_PDCP_DelayResult(qci1), cr_UL_PDCP_DelayResult(qci2), *)}

    timer t_ReportInterval := 5.632; // 5120ms + 10%

    var Frequency_IE_Type v_Frequency_IE_f1;

    var ChannelBandwidthDependency_Type v_ChBandDependency_Cell1;

    var Dl_Bandwidth_Type v_DlBandwidth;

    // Init variables

    f_EUTRA_Init(c1);

    v_Sr_PUCCH_ResourceIndex := f_EUTRA_CellInfo_GetSR_PUCCH_ResourceIndex(eutra_Cell1)

    v_SR_ConfigIndex := f_EUTRA_CellInfo_GetSR_ConfigIndex (eutra_Cell1);

    v_Tft_Dedicated1 := f_EUTRA_Get_508_TrafficFlowTemplate(1, tsc_EpsDedicatedBearerId);   // Dedicated EPS bearer context #1

    v_Tft_Default2 := f_EUTRA_Get_508_TrafficFlowTemplate(2, tsc_EpsDefaultBearerId2);      // Default EPS bearer #2

    v_Tft_Dedicated3 := f_EUTRA_Get_508_TrafficFlowTemplate(3, tsc_EpsDedicatedBearerId3);  // Dedicated EPS bearer context #3

    v_Tft_List := {v_Tft_Dedicated1, v_Tft_Default2, v_Tft_Dedicated3};

    v_Frequency_IE_f1 := f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1);

    v_ChBandDependency_Cell1 := f_EUTRA_BandDependentParam(v_Frequency_IE_f1.DL_ChBandwidth, v_Frequency_IE_f1.UL_ChBandwidth);
    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -70);

    // Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);    

    f_EUTRA_SetCellPower(eutra_Cell1, tsc_ServingCellRS_EPRE);

    f_EUTRA_SwitchOnUEandStartIP(eutra_Cell1);
    // The UE is in state Registered, Idle Mode (State 2) according to [18].

    f_EUTRA_InitialRegistration(eutra_Cell1, PREAMBLE, STATE2A_TESTLOOP_ModeB, NORMAL);

    f_EUTRA_TestBody_Set (true);

    //@siclog "Step 0" siclog@

    /* The SS performs the generic procedure to get UE in Loopback Activated (state 4) with condition UE TEST LOOP MODE B with a Reference default EPS bearer context #1 and

    Reference dedicated EPS bearer context #1 and Reference default EPS bearer #2 and Reference dedicated EPS bearer context #3 according to [18] on Cell A */

    f_EUTRA_GenericRbEst(eutra_Cell1, 3, 0, v_Qos_List, v_Tft_List);

    f_EUTRA_CloseUE_TestLoopModeB(eutra_Cell1, '00'O);

    //@siclog "Step 1-2" siclog@

    //The SS transmits an  RRCConnectionReconfiguration message including MeasConfig to setup UL-DelayConfig 
   f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(v_RRC_TI,

                                                                                                     cs_MeasConfig(omit,

                                                                                                                   cs_MeasObjectToAddModList_1Entry(tsc_IdMeasObject_f1,

                                                                                                                   cs_MeasObject_measObjectEUTRA_Common(f_EUTRA_ARFCN_Value(v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq_r9), v_ChBandDependency_Cell1.AllowedMeasBandwidth)),

                                                                                                                   omit,

                                                                                                                   cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_PeriodicalCsiRs,

                                                                                                                                                      cs_ReportConfigToAddMod_ReportConfigEUTRA_MeasCSI_RS_Id(ms5120, r64, cs_UL_DelayConfig_Setup_r13(ms750), omit)),

                                                                                                                   omit,                                                                                                                                                      

                                                                                                                   cs_MeasIdToAddModList_1Entry(1, tsc_IdMeasObject_f1, tsc_IdReportConfig_PeriodicalCsiRs ),-,-,-,-,-,

                                                                                                                   f_ReturnMeasObjectToAddModList_v9e0(v_Frequency_IE_f1.UL_DL_Earfcn.dl_CarrierFreq_r9)),

                                                                                                                   cds_PhysicalConfigDedicated_DSR_Transmax(v_Sr_PUCCH_ResourceIndex, v_SR_ConfigIndex)));
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 2");

    v_PdnType := f_EUTRA_MobileInfo_GetAssignedPdnType();

    v_UseIPv4 := (v_PdnType == tsc_PdnType_IPv4);  /* @sic R5-112600: clarification to conditions for IP address configuration sic@ */

    v_IP_Address1_NW := f_LoopbackModeB_IP_Address_NW(v_UseIPv4);      /* @sic R5s120470 sic@ */

    v_UdpPacket_DRB1 := f_IPv4IPv6_AnyUdpPacket(v_IP_Address1_NW, v_IP_Address1_NW, 31160, 61000);

    v_UdpPacket_DRB3 := f_IPv4IPv6_AnyUdpPacket(v_IP_Address1_NW, v_IP_Address1_NW, 60352, 60352);
    //@siclog "Step 3" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #1

    //DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS1, cs_TimingInfo_Now, v_PDCP_SDUList));

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS1, cs_TimingInfo_Now, { v_UdpPacket_DRB1 }));
    //@siclog "Step 4" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #3

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS3, cs_TimingInfo_Now, { v_UdpPacket_DRB3 }));
    //@siclog "Step 5" siclog@

    //SS provides sufficient UL grants to ensure UE can loopback IP packets without delay and waits for duration indicated in the IE reportInterval

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);

    //@siclog "Step 6" siclog@

    //Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2

    t_ReportInterval.start;

    alt{

        [] SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MeasurementReport_CheckAny))-> value v_MeasReport

        {

            if(ispresent(v_MeasReport.Signalling.Rrc.Dcch.message_.c1.measurementReport.criticalExtensions.c1.measurementReport_r8.measResults.measResultCSI_RS_List_r12))

            {

                t_ReportInterval.stop;

                f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 6: UE transmits a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2");

            }

            else

            {

                repeat;

            }
        }

        [] t_ReportInterval.timeout

        {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 6: UE does not transmit a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2");

        }

        [] DRB.receive (cr_DRB_COMMON_IND_Any)

        {

            repeat;        

        }
    }

    //@siclog "Step 7" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #1

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS1, cs_TimingInfo_Now, { v_UdpPacket_DRB1 }));
    //@siclog "Step 8" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #3

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS3, cs_TimingInfo_Now, { v_UdpPacket_DRB3 }));
    //@siclog "Step 9" siclog@

    //SS doesn't provides UL grants to UE and waits for duration indicated in the IE reportInterval

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

    f_Delay(0.750);

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);
    //@siclog "Step 10" siclog@

    //Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2

      t_ReportInterval.start;

    alt{

        [] SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cdr_MeasurementReport_UL_PDPC_DelayResults(1,-,-,-,-,v_MeasList)))

        {

            t_ReportInterval.stop;

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 10: UE transmits a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and QCI 2");

        }

        [] t_ReportInterval.timeout

        {

            f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 10: UE transmits a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2");

        }

        [] DRB.receive (cr_DRB_COMMON_IND_Any)
        {

                repeat;            

        }
    }

    //@siclog "Step 11-12" siclog@

    //The SS transmits an RRCConnectionReconfiguration message including MeasConfig to release the UL-DelayConfig measurement

    f_EUTRA_RRCConnectionReconfiguration_MeasConfig(eutra_Cell1,

                                                    cs_508_RRCConnectionReconfiguration_ConditionMEAS(v_RRC_TI,

                                                                                                     cs_MeasConfig(omit,

                                                                                                                   omit,

                                                                                                                   omit,

                                                                                                                   cs_ReportConfigToAddModList_1Entry(tsc_IdReportConfig_PeriodicalCsiRs,

                                                                                                                                                      cs_ReportConfigToAddMod_ReportConfigEUTRA_MeasCSI_RS_Id(ms10240, r1, cs_UL_DelayConfig_Release_r13)),

                                                                                                                   omit,

                                                                                                                   omit)));

    //@siclog "Step 13" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #1

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS1, cs_TimingInfo_Now, { v_UdpPacket_DRB1 }));

    //@siclog "Step 14" siclog@

    //The SS transmits an IP Packet on the data radio bearer associated with the dedicated EPS bearer contex #3

    DRB.send(cas_DRB_COMMON_REQ_PDCP_SDU(eutra_Cell1,  v_RbId_DedicatedEPS3, cs_TimingInfo_Now, { v_UdpPacket_DRB3 }));

    //@siclog "Step 15" siclog@

    //SS doesn't provides UL grants to UE and waits for duration indicated in the IE reportInterval

    f_EUTRA_StopULGrantTransmission_PUCCH_Sync(eutra_Cell1, cs_TimingInfo_Now);

    f_Delay(0.750);

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);  

    //@siclog "Step 16" siclog@

    //Check: Does the UE transmit a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and QCI 2

    t_ReportInterval.start;

    alt{

        [] SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_MeasurementReport_CheckAny))-> value v_MeasReport

        {

            if(ispresent(v_MeasReport.Signalling.Rrc.Dcch.message_.c1.measurementReport.criticalExtensions.c1.measurementReport_r8.measResults.measResultCSI_RS_List_r12))

            {

                t_ReportInterval.stop;

                f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 16: UE transmits a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2");

            }

        }

        [] t_ReportInterval.timeout

        {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 8.6.1.3 Step 16: UE does not transmit a MeasurementReport message containing UL-PDCP-DelayResultList for QCI 1 and/or QCI 2");

        }

        [] DRB.receive (cr_DRB_COMMON_IND_Any)
        {

            repeat;        

        }
    }

    f_EUTRA_TestBody_Set (false);

    // Postamble

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED);

  }

}
Change 2
	Function name
	template (value) DL_DCCH_Message cs_508_RRCConnectionReconfiguration_ConditionMEAS

	Reason for change
	The default value of Maximum number of SR transmission count is 4. But in this case it needs to increased to maximum value 64.

	Summary of change
	Parameter to alter the value of Dsr_TransMax as been added in cs_508_RRCConnectionReconfiguration_ConditionMEAS.

	TTCN module
	LTE_A_PRO_IWD_18wk37\Common\EUTRA\ EUTRA_Measurement_Templates.ttcn

	MCC160 Comment
	


Before change

   template (value) DL_DCCH_Message cs_508_RRCConnectionReconfiguration_ConditionMEAS(RRC_TransactionIdentifier p_RRC_TI,

                                                                                     template (value) MeasConfig p_MeasConfig ) :=
    /* @status    APPROVED (LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_IRAT) */

    cs_RRCConnectionReconfiguration_Common( p_RRC_TI, p_MeasConfig, omit, omit, omit );
After change
  template (value) DL_DCCH_Message cs_508_RRCConnectionReconfiguration_ConditionMEAS(RRC_TransactionIdentifier p_RRC_TI,

                                                                                     template (value) MeasConfig p_MeasConfig,

                                                                                     template (omit) PhysicalConfigDedicated p_PhysicalConfigDedicated := omit ) :=

    /* @status    APPROVED (LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_IRAT) */

    cs_RRCConnectionReconfiguration_Common( p_RRC_TI, p_MeasConfig, omit, cs_RadioResourceConfigDedicated_MacPhy (-, -, p_PhysicalConfigDedicated), omit );
5 Execution Log Files

INTEL 7660
The INTEL 7660 passed this test case on Anite Conformance Toolset Solution in LTE FDD band 4. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test Case Execution log file: 
TC_8_6_1_3_FDD.html
6 References

	[1]
	R5s180588: Supporting information for agreement of test case 8.6.1.3. This archive comprises html and xml format execution log files and PICS/PIXIT settings file respectively.

	
	

	
	


